Open-File Report

PETAL: PEnetration Testing And Liquefaction, An Interactive Computer Program

Albert T. F. Chen

U.S. Geological Survey
Menlo Park, California 94025

Open-File Report 84-290

March, 1984

INTRODUCTION

Recent work on evaluation of liquefaction potential using field performance data by Seed and Others (1982, 1983) has led to a simple yet effective procedure through which the liquefaction potential of given sand deposits can be determined with minimum amount of testing efforts from the laboratory. This procedure uses standard penetration test (SPT) or cone penetration test (CPT) data as input and applies to sands and to silty sands.

The report describes the computer program PETAL (for PEnetration Testing And Liquefaction) which performs various steps of computations required in the simplified procedure and determines the factor of safety against liquefaction for given sand layers of a site under specific seismic shaking conditions. PETAL was coded in FORTRAN and programmed to run in an interactive mode with a VAX 11/780 computer and a VERSATEC (Model 1200) plotter. The storage requirement to execute PETAL is less than 23-K bytes thus the program can be easily modified to run on many personal computers.

In this report, the criteria on which PETAL is based are presented. In addition, the program listing is included and the use of the program is illustrated. Justifications and further details on the technical aspects of the procedure may be found in the paper by Seed and Others (1983).

GENERAL DESCRIPTION

Site Description. - The required input for the description of a site are the total number of layers, the depth from the the surface to the lower boundary of each layer (in ft), the bulk density of each layer (in pcf), and the depth of the ground water table (in ft). These information are necessary for computing the total and the effective stresses later in the program. The seismic event is specified by the magnitude of the earthquake, M, and the estimated maximum horizontal ground acceleration at the site, a_{max} (in g). It is assumed that M is the controlling parameter on the duration of shaking and therefore the number of stress cycles, and that a_{max} is directly related to the maximum shear stress experienced at a given depth.

Chart for Evaluation of Liquefaction Potential. - The chart for M=7.5 earthquake given by Seed and Idriss (1982, p.107, Figure 54) was digitized and stored in PETAL. The digitized curve is reproduced in Figure 1. Different charts for earthquakes with magnitudes other than 7.5 are obtained by mutiplying the boundary curve with an appropriate scaling factor, FAC. Subroutine GETFAC first determines the number of representative stress cycles, N_{sc}, from an assigned earthquake magnitude according to the relation shown in Figure 2. Then, using N_{sc}, the same subroutine computes FAC according to the relationship shown in Figure 3.

Additional Input Required. - After the site and the seismic event have been specified, PETAL is ready to evaluate the liquefaction potential of the sand layers at that site. Each sand layer is specified by its depth, z (in ft), standard penetration resistance (in blow count) or cone penetration tip resistance, and relative density. The cone penetration tip resistance, q_c (in kg/cm^2) should be entered with a minus sign for the program to identify the correct type of input. If the sand is considered silty (with $D_{50}<0.15$ mm), the relative density should be entered with a minus sign for the same reason. Input on relative density and grain size (D_{50}) are needed for determining the correct normalized penetration resistance later in the program. Once these input are entered from the computer terminal, PETAL



Figure 1. -- Chart for evaluation of liquefaction potential for 7.5 magnitude earthquakes



Figure 2. -- Relationship between number of representative cycles at 0.65 τ_{max} and earthquake magnitude



Figure 3. -- Relationship between scaling factor and number of representative cycles at $0.65\tau_{max}$

proceeds to compute the average cyclic stress ratio, to make necessary conversions and corrections (to be described below), and finally to determine the factor of safety against liquefaction. Unless otherwise specified, the program continues to repeat the same tasks for other sand layers at the same site.

Average Cyclic Stress Ratio. - The stress ratio developed in the field during the assigned earthquake event is computed by:

$$\frac{(\tau)_{ave}}{\sigma_o'} = 0.65 \frac{a_{max}}{g} \frac{\sigma_o}{\sigma_o'} \cdot r_d \quad (1)$$

in which a_{max} = input maximum horizontal acceleration; g = gravitational acceleration; σ_o = total overburden pressure on sand layer under consideration; σ_o' = intial effective overburden pressure on sand layer under consideration; and r_d = stress reduction factor. r_d is assumed to vary with depth in the manner shown in Figure 4 and is determined in Petal by interpolating the digitized values stored for this curve.

Correction for Shallow Depth. - If the depth of the sand layer under consideration is less than 10 ft, the standard penetration blow count is multiplied by 0.75 to allow energy loss in the drive rods.

Correlations between SPT and CPT Test Data.- For clean sands, CPT data is converted to an equivalent SPT blow count, N, by dividing q_c by a factor of 4.5. For silty sands, the factor is 4.0.

Normalized Penetration Resistance. - A measured or converted (standard) penetration resistance, N, is modified with respect to an effective overburden pressure of 1 ton/ft^2. The normalized penetration resistance (blow count),



Figure 4. -- Stress reduction factor as a function of depth

N_1, is determined from the relation,

$$N_1 = C_N \cdot N \tag{2}$$

in which C_N is a function of the effective overburden pressure at the depth of the sand layer under consideration. In PETAL, C_N is determined by interpolating the digitized values of the relations shown in Figure 5.

Corrections for Silty Sands. - When the sand layer under consideration has D_{50}<0.15mm, it is considered silty and its modified penetration resistance, N_1 is increased by 7.5 to correct for grain size effects.

Factor of Safety. - The final modified penetration resistance (after required correction), N_1, and the average cyclic stress ratio computed from Eq. (2) represents a point on the chart for liquefaction evaluation developed for the input earthquake magnitude. The chart is similar to Figure 1 which was intended for M=7.5 earthquakes. If the point is located above the curve, occurence of liquefaction is suggested. The factor of safety against liquefaction is defined as

$$F.S. = \frac{(\tau/\sigma_o')_l}{(\tau/\sigma_o')_{in\ situ}} \tag{3}$$

in which $(\tau/\sigma_o')_{in\ situ}$ is the average cyclic stress ratio in situ (from Eq. (2)) and $(\tau/\sigma_o')_l$ is the cyclic stress ratio to cause liquefaction at the computed N_1 for the sand layer under consideration. The latter stress ratio is determined by interpolation from the developed chart in PETAL.

Plotfiles. - When computations for specific sand layers of the given site are completed, the user is given the the option to have plotfiles generated



Figure 5. -- Conversion factor as a function of the effective overburden pressure and relative density

and the plot made. In addition, computed results are automatically stored on a file assigned to I/O Unit 16 (FOR016 for the VAX System) for future use.

SAMPLE RUN

For a demonstration run, consider a site consisting of two layers with their depths and densities given below:

Layer 1	15 ft	100 pcf
Layer 2	50 ft	110 pcf

The water table is at 10 ft. The earthquake magnitude is 6.6 and the maximum ground acceleration is 0.40 g.

PETAL is applied to evaluate the liquefaction potential for sand layers specified below:

DEPTH	TYPE	SPT, N	CPT, q_c	RELATIVE DENSITY
8.0	sand	6.0		40%
15.0	silty sand	10.0		40%
20.0	silty sand		100 kg/cm^2	60%

All input are to be entered from the keyboard of the terminal. The following is a reproduction of what appeared on the screen/printer during this computer run. For distinction, input from the keyboard are printed in light italic. The plot and the output file from I/O UNIT 16 produced from this run are also included.

(REPRODUCTION OF THE INTERACTIVE SESSION DURING THE DEMONSTRATION RUN)

to assess liquefaction potential from spt data
all inputs should be in u.s. customary units
such as LBS or FT, except for CPT tip resistance
which should be in KG/SQ. CM

enter title of this run in 72 capitalized letters
PETAL DEMO RUN, M=6.6, AMAX=0.4, 2/22/84

site description: enter no. of layers (<10)
2

enter depth(ft) and density(pcf) with decimal of layer 1
15.0, 100.0

enter depth(ft) and density(pcf) with decimal of layer 2
50.0, 110.0

enter depth of ground water table in ft (13.0)
10.0

enter equake mag. and max acc (g) -- 7.5, 0.25
6.6, 0.4

enter depth (ft) & spt blow count, i.e. 18.0, 23.0, HOWEVER,
if cpt, enter depth and Qc in kg/sq.cm with a MINUS sign, i.e. 18.0, -92.0
8.0, 6.0

enter relative density (0.4 for 40%), with a MINUS sign if sand is silty
0.4

depth above ground water, no liquefaction
enter an integer>0 for a new depth to consider
9

enter depth (ft) & spt blow count, i.e. 18.0, 23.0, HOWEVER,
if cpt, enter depth and Qc in kg/sq.cm with a MINUS sign, i.e. 18.0, -92.0
15.0, 10.0

enter relative density (0.4 for 40%), with a MINUS sign if sand is silty
-0.4

depth of 15.0 ft is in layer 1 eff. & total stress = 1188.0 1500.0

average stress ratio = 0.32

value of cn = 1.29 modified blow count = 20.4

stress ratio insitu = 0.318 required to cause liq. = 0.249
factor of safety = 0.78

enter an integer>0 for a new depth to consider
9

```
 enter depth (ft) & spt blow count, i.e. 18.0, 23.0,  HOWEVER,
 if cpt, enter depth and Qc in kg/sq.cm with a MINUS sign, i.e. 18.0, -92.0
   20.0, -100.0

 enter relative density (0.4 for 40%), with a MINUS sign if sand is silty
   -0.6

depth of 20.0 ft is in layer 2  eff. & total stress =     1426.0    2050.0

average stress ratio = 0.36

value of cn = 1.18  modified blow count = 37.1

stress ratio insitu = 0.356 required to cause liq. = 0.603
factor of safety = 1.69

 enter an integer>0 for a new depth to consider
  0


 enter integer >0 if plotfile is needed
  9
```

(REPRODUCTION OF THE PLOT FROM THE DEMONSTRATION RUN)



```
PETAL DEMO RUN, M=6.6, AMAX=0.4, 2/22/84

the site consists of  2 layers w/ depths & dens:
                      1      15.0 (ft)          100.0 (pcf)
                      2      50.0 (ft)          110.0 (pcf)

input eq. mag.= 6.60   max. acc. = 0.40 g and ground water table depth =  10.0 ft.

count   depth   mea.spt   ef.stress   tot.stress   insitu    mod.spt   liquef.   factor              spt/cpt
         (ft)   (blows)     (psf)       (psf)     st.ratio   (blows)  st.ratio   safety   remarks    input

  1       8.0       4.5         0.0          0.0    0.000        0.0     0.000    0.00                  6.0
  2      15.0      10.0      1188.0       1500.0    0.318       20.4     0.249    0.78        silt     10.0
  3      20.0      25.0      1426.0       2050.0    0.356       37.1     0.603    1.69    cpt silt   -100.0
```

(OUTPUT FILE ON UNIT 16)

ACKNOWLEDGEMENT

Digitization of the empirical curves originally given by Seed and Idriss (1982) was provided by R. Mark of U.S. Geological Survey.

REFERENCES CITED

Seed, H. B., and Idriss, I. M., 1982, Ground Motions and Soil Liquefaction during Earthquakes: EERI Monograph Series, Earthquake Engineering Research Institute

Seed, H. B., Idriss, I. M., and Arrango, I., 1983, Evaluation of liquefaction potential using field performance data: Journal of Geotechnical Engineering, ASCE, Vol. 109, No. 3, March, pp.458-482

PROGRAM LISTING

If the user has access to the VAX 11/780 Computer of the Office of Earthquakes, Volcanoes, and Engineering, USGS in Menlo Park, California, he can find PETAL residing in the directory:

PUB1:[CHEN.LIQ]

and can simply execute the program by typing the command:

@petal

to activate the command procedure, petal.com, which will link the designated files, run the program, and generate plots if specified. In this case, file PETAL.OUT is automatically assigned to I/O Unit 16.

Listings of petal.com, petal and its subroutines are reproduced in the following pages.

```
c         PETAL: PEnetration Testing And Liquefaction
c
c                program to estimate liquefaction potentials from
c                SPT or CPT data
c
c         ref: seed, journal of geotechnical engineering, asce,
c                   vol. 109, no. 3, march, 1983
c
c         by a. chen, oeve, usgs, 2/84
c         digitization of empirical curves were provided by r. mark
c
      dimension dref(9), rd(9), sv8(16), cn8(16), sv4(16), cn4(16),
     & xf(20), yf(20), depth(9),rmk(4), den(9), th(9)
      common /blka/x(9),y(9),xn(9),yt(9),title(18),resu(11,15)
c
      data rmk/'    ',' cpt','silt','shlw'/
      data rd/1.0,0.9794,0.9668,0.9478,0.9346,0.9189,0.9009,
     &        0.8709,0.40/
      data dref/0.0,11.825,15.469,21.643,27.268,31.752,34.813,
     &          39.535,100.0/
      data xn/5.288,11.014,15.308,20.702,26.094,29.823,31.468,33.426,
     &        34.714/
      data yt/0.05333,0.1133,0.1588,0.2166,0.2765,0.3297,0.3529,0.3949,
     &        0.4379/
      data sv8/0.7732,0.9447,1.2934,1.7221,1.9845,2.2949,2.6744,3.1689,
     &         3.5984,4.1400,4.7297,5.3664,6.1172,7.2153,8.1312,9.0241/
      data cn8/1.5965,1.4295,1.2288,1.0780,1.0114,0.9536,0.8951,0.8357,
     &         0.7952,0.7400,0.6936,0.6513,0.6035,0.5619,0.5310,0.5003/
      data sv4/0.7732,0.9447,1.2934,1.7221,1.9845,2.1597,2.5362,2.9828,
     &         3.4533,4.0370,4.5796,5.1473,5.8070,6.7640,7.7940,8.7560/
      data cn4/1.5965,1.4295,1.2288,1.0780,1.0114,0.9685,0.8963,0.8281,
     &         0.7643,0.6903,0.6397,0.5980,0.5556,0.5014,0.4649,0.4337/
c
c         format statements
c
    2 format(18a4)
    4 format('    ')
    6 format(' enter depth(ft) and density(pcf) with decimal of layer',
     & i3)
    8 format(' depth of',f5.1,' ft is in layer',i2,
     & '   eff. & total stress =',2f10.1/)
   10 format(' average stress ratio =',f5.2/)
   12 format(' value of cn =',f5.2,'  modified blow count =',
     & f5.1/)
   14 format(' stress ratio insitu =',f6.3,' required to cause liq. =',
     & f6.3/' factor of safety =',f5.2/)
   16 format(/' the site consists of',i3,' layers w/ depths & dens:')
   18 format(20x,i4,f10.1,' (ft)',f15.1,' (pcf)')
   20 format(/' input eq. mag.=',f5.2,'   max. acc. =',f5.2,
     & ' g and ground water table depth =',f6.1,' ft.')
   22 format(//' count   depth   mea.spt   ef.stress   tot.stress   ',
     & 'insitu   mod.spt   liquef.   factor',14x,'spt/cpt'/
     & 10x,'(ft)   (blows)',4x,
```

```
     & ' (psf)       (psf)    st.ratio   (blows)   st.ratio  safety',
     & '   remarks   input'/)
   24 format(i4.2f10.1 2f12.1,f10.3,f10.1,f10.3,f8.2,3x,a4,1x,a4,f8.1)
c
      write(6,4)
      write(6,4)
      write(6,4)
      write(6,*) ' to assess liquefaction potential from spt data'
      write(6,*) ' all inputs should be in u.s. customary units '
      write(6,*) ' such as LBS or FT, except for CPT tip resistance'
      write(6,*) ' which should be in KG/SQ. CM'
      write(6,4)
      write(6,*) ' enter title of this run in 72 capitalized letters'
      write(6,4)
      read(5,2) title
      write(6,4)
      write(6,*) ' site description: enter no. of layers (<10)'
      write(6,4)
      read*, nlayer
      do  40 i=1,nlayer
      write(6,6) i
      write(6,4)
      read*, depth(i), den(i)
   40 continue
      th(1) = depth(1)
      do  60 i=2,nlayer
      th(i) = depth(i) - depth(i-1)
   60 continue
      write(6,4)
      write(6,*) ' enter depth of ground water table in ft (13.0)'
      write(6,4)
      read*, zgw
      write(6,4)
      write(6,*) ' enter equake mag. and max acc (g) -- 7.5, 0.25'
      write(6,4)
      read*, eqm, amax
      write(6,4)
c
      call getfac(eqm,fac)
c
c        to establish reference stress-ratio vs n1 curve
c
      do  80 i=1,9
      yt(i) = yt(i)*fac
   80 continue
c
      ic = 0
  100 ic = ic+1
      itype = 0
      isilt = 0
      xcpt = 1.0
      write(6,*) ' enter depth (ft) & spt blow count, i.e. 18.0, 23.0
     &,  HOWEVER,'
      write(6,*) ' if cpt, enter depth and Qc in kg/sq.cm with a MINUS
```

```
     $ sign, i.e. 18.0, -92.0'
      write(6,4)
      read*, z, bc
      write(6,4)
      if(z .gt. 40.) write(6,*) ' warning: depth exceeds 40 ft'
      write(6,*) ' enter relative density (0.4 for 40%), with a MINUS
     & sign if sand is silty'
      write(6,4)
      read*, rden
      write(6,4)
c
c         make corrections for cpt input (itype=1); silty sands if
c         D50<0.15mm (isilt=1); and shallow depth
c
      resu(11,ic) = bc
      if(bc .ge. 0.0) go to 120
      itype = 1
      bc = abs(bc)
      xcpt = 4.5
  120 if(rden .ge. 0.0) go to 140
      isilt = 1
      rden = abs(rden)
      if(itype .eq. 1) xcpt=4.0
  140 bc = bc/xcpt
      if(z .lt. 10. .and. itype .eq. 0) bc=0.75*bc
c
c        compute total and effective stress at the given depth
c
      if(z .gt. zgw) go to 220
      write(6,*) ' depth above ground water, no liquefaction'
      resu(1,ic) = z
      resu(2,ic) = bc
      do 210 j=3,8
      resu(j,ic) = 0.0
  210 continue
      resu(9,ic) = rmk(1)
      resu(10,ic) = rmk(1)
      go to 800
c
  220 continue
      sum1 = 0.0
      sum2 = 0.0
      j = 0
      do 240 loop=1,nlayer
      j = j+1
      if(depth(j) .ge. zgw) go to 250
      sum1 = sum1 + th(j)*den(j)
      sum2 = sum2 + th(j)*den(j)
  240 continue
  250 continue
      idry = j
      if(idry .gt. 1) go to 280
      if(z .gt. depth(1)) go to 260
c
```

```
c        z, zgw both in layer 1
      sum1 = zgw*den(1) + (z-zgw)*(den(1)-62.4)
      sum2 = z*den(1)
      go to 400
c
  260 sum1 = zgw*den(1) + (depth(1)-zgw)*(den(1)-62.4)
      sum2 = depth(1)*den(1)
      go to 320
  280 if(z .gt. depth(idry)) go to 300
      sum1 = sum1 + (zgw-depth(idry-1))*den(idry)
     &            + (z-zgw)*(den(idry)-62.4)
      sum2 = sum2 + (z-depth(idry-1))*den(idry)
      go to 400
  300 sum1 = sum1 + (zgw-depth(idry-1))*den(idry)
     &            + (depth(idry)-zgw)*(den(idry)-62.4)
      sum2 = sum2 + th(idry)*den(idry)
  320 continue
      do 340 loop=idry,nlayer
      j = j+1
      if(depth(j) .gt. z) go to 360
      sum1 = sum1 + th(j)*(den(j)-62.4)
      sum2 = sum2 + th(j)*den(j)
  340 continue
  360 sum1 = sum1 + (z-depth(j-1))*(den(j)-62.4)
      sum2 = sum2 + (z-depth(j-1))*den(j)
  400 continue
      write(6,8) z, j, sum1, sum2
c
c        to determine stress reduction factor rd & ave stress-ratio
c
      j = 1
      do 420 loop=1,8
      j = j+1
      if(dref(j) .gt. z) go to 440
  420 continue
  440 fac1 = rd(j-1) + (z-dref(j-1))*(rd(j)-rd(j-1))/(dref(j)-dref(j-1))
      atau = 0.65*fac1*amax*sum2
      taur = atau/sum1
      write(6,10) taur
c
c        to determine modified penetration resistance
c
      ysig = sum1/1000.0
      if(rden .ge. 0.60) go to 480
      do 460 i=1,16
      xf(i) = cn4(i)
      yf(i) = sv4(i)
  460 continue
      go to 500
  480 do 490 i=1,16
      xf(i) = cn8(i)
      yf(i) = sv8(i)
  490 continue
  500 continue
```

```
      if(ysig .gt. yf(1)) go to 520
      write(6,*) ' vert. eff. stress out of range, assume cn=1.8'
      cn = 1.8
      go to 580
  520 continue
      j = 1
      do 540 loop=1,15
      j = j+1
      if(yf(j) .gt. ysig) go to 560
  540 continue
  560 cn = xf(j-1) + (xf(j)-xf(j-1))*(ysig-yf(j-1))/(yf(j)-yf(j-1))
  580 continue
      bcmod = bc*cn
      if(isilt .eq. 1) bcmod=bcmod+7.5
      write(6,12) cn,bcmod
c
c        to determine stress ratio at 100% pore pressure ratio
c
      j = 1
      do 600 loop=1,8
      j = j+1
      if(xn(j) .gt. bcmod) go to 620
  600 continue
c
  620 ratiof = yt(j-1) + (yt(j)-yt(j-1))*(bcmod-xn(j-1))/(xn(j)-xn(j-1))
      sf = ratiof/taur
      write(6,14) taur, ratiof, sf
c
c        up to this point, computation for liq pont at a given depth is
c        done.  next is to store results for further use and see if
c        continuation for another depth is requested.
c
      resu(1,ic) = z
      resu(2,ic) = bc
      resu(3,ic) = sum1
      resu(4,ic) = sum2
      resu(5,ic) = taur
      resu(6,ic) = bcmod
      resu(7,ic) = ratiof
      resu(8,ic) = sf
      resu(9,ic) = rmk(1)
      resu(10,ic) = rmk(1)
      if(itype .eq. 1) resu(9,ic)=rmk(2)
      if(isilt .eq. 1) resu(10,ic)=rmk(3)
      if(itype .eq. 0 .and. z .lt. 10.) resu(9,ic)=rmk(4)
c
  800 continue
      write(6,*) ' enter an integer>0 for a new depth to consider'
      write(6,4)
      read*, icont
      if(icont .gt. 0) go to 100
c
c        save results on designated file for printer output
c
```

```
      write(16,2) title
      write(16,16) nlayer
      write(16,18) ((i,depth(i),den(i)),i=1,nlayer)
      write(16,20) eqm, amax, zgw
      write(16,22)
      do 850 i=1,ic
      write(16,24) i,(resu(j,i),j=1,11)
  850 continue
c
c        to check if plot of results is requested
c
      write(6,4)
      write(6,4)
      write(6,*) ' enter integer>0 if plotfile is needed'
      write(6,4)
      read*, iplt
      if (iplt .le. 0) go to 900
      call pltliq(ic)
  900 continue
c
      stop
      end




      subroutine pltliq(ic)
c
c         to be used with program petal for plotting liquefaction
c         potential of a given site. 2/84 by a. chen
c
      dimension xamp(5), yamp(6), xs(15), ys(15)
      common /blka/x(9), y(9), xn(9), yt(9), title(18), resu(11,15)
      data xamp/'   0',' 10',' 20',' 30',' 40'/
      data yamp/'   0',' 0.1',' 0.2',' 0.3',' 0.4',' 0.5'/
c
c         plot grid
c
      call plots(0,0,0)
      call plot(4.,1.5,3)
      call plot(8.,1.5,2)
      call plot(8.,6.5,2)
      call plot(4.,6.5,2)
      call plot(4.,1.5,2)
c
      xa = 4.0
      ya = 1.5
      yb = 1.65
      yc = 6.5
      yd = 6.35
      do 50 i=1,3
      xa = xa + 1.0
      call plot(xa,ya,3)
```

```
      call plot(xa,yb,2)
      call plot(xa,yc,3)
      call plot(xa,yd,2)
   50 continue
      ya = 1.5
      xa = 4.0
      xb = 4.15
      xc = 8.0
      xd = 7.85
      do 100 i=1,4
      ya = ya + 1.0
      call plot(xa,ya,3)
      call plot(xb,ya,2)
      call plot(xc,ya,3)
      call plot(xd,ya,2)
  100 continue
      call plot(4.,1.5,3)
c
      call symbol(3.3,0.6,0.18,31hMODIFIED PENETRATION RESISTANCE,0.,31)
      xa = 2.5
      ya = 1.2
      do 150 i=1,5
      xa = xa + 1.0
      call symbol(xa,ya,0.18,xamp(i),0.,4)
  150 continue
      ya = 0.42
      xa = 3.2
      do 200 i=1,6
      ya = ya + 1.0
      call symbol(xa,ya,0.18,yamp(i),0.,4)
  200 continue
      call symbol(3.0,2.42,0.18,19hCYCLIC STRESS RATIO,90.0,19)
      call symbol(3.0,9.5,0.06,title,0.,72)
c
c        plot ref n1 vs. stress ratio curve for given eqake magnitude
c
      xa = 4.0 + 0.1*xn(1)
      ya = 1.5 + 10.0*yt(1)
      call plot(xa,ya,3)
      do 250 i=2,9
      xa = 4.0 + 0.1*xn(i)
      ya = 1.5 + 10.0*yt(i)
      call plot(xa,ya,2)
  250 continue
      call plot(4.,1.5,3)
c
c         plot results at given depths
      do 300 i=1,ic
      xs(i) = 4.0 + 0.1*resu(6,i)
      ys(i) = 1.5 + 10.0*resu(5,i)
  300 continue
      xs(ic+1) = 0.0
      xs(ic+2) = 1.0
      ys(ic+1) = 0.0
```

```
      ys(ic+2) = 1.0
c
      call line(xs,ys,ic,1,-1,11)
c
      call plot (8.,1.5,999)
      return
      end





      subroutine getfac(eqm,fac)
c
c        subroutine to compute scaling factor, fac, for a
c        given earthquake magnitude, eqm, to establish the
c        reference liquefaction potential curve --
c        stress ratio versus modified penetration blow count
c
      dimension sy(5),qx(5),cy(5)
      data sy/1.6,1.32,1.13,1.0,0.89/
      data qx/5.25,6.0,6.75,7.5,8.5/
      data cy/3.0,6.0,10.0,15.0,26.0/
c
      do 100 i=1,4
      if(eqm .le. qx(i+1)) go to 120
  100 continue
  120 cyn=cy(i)+(eqm-qx(i))*(cy(i+1)-cy(i))/(qx(i+1)-qx(i))
      do 140 i=1,4
      if(cyn .le. cy(i+1)) go to 160
  140 continue
  160 fac=sy(i)+(cyn-cy(i))*(sy(i+1)-sy(i))/(cy(i+1)-cy(i))
      return
      end
```

(PETAL.COM)

```
$ define sys$input sys$command:
$ assign petal.out for016
$ run petal
$ inquire p1 "enter yes if plot needs plotted"
$ if p1 .nes. "YES" then goto ok
$ mcr rasm
$ write sys$output " del *.plv;*"
$ ok:
$ type petal.out
```